

## 2.3 FreeFEM++入門

### 2.3.1 FreeFEM++とは (文献 [4] からの引用)

FreeFEM プロジェクトはパリ第6大学のO. ピロノ (Pironneau) によってはじめられ, その後, F. エヒト (Hecht, パリ第6大学), 大塚厚二 (広島国際学院大学) らの協力のもとに進められている.

### 2.3.2 インストール

FreeFEM++は <http://www.freefem.org/> からダウンロードできる. また, FreeFEM++用統合環境 FreeFEM++-cs も用意されているので, その統合環境を <http://www.ann.jussieu.fr/~lehyaric/ffcs/index.htm> からダウンロードすることを勧める (とりあえずは, FreeFEM++-cs をインストールするだけ良い.)

### 2.3.3 三角形分割

図 2.15 のような  $W$  型領域を考える. この領域の三角形分割を生成する. プログラムは下記の `w-triangulation.edp`<sup>5</sup> のようになる. 生成結果は図 2.16 のようになる. その三角形分割データは `w-triangulation.msh` に保存される (詳しくは .2.3.4 参照). `w-triangulation.edp` のプログラムでは, `border` コマンドによって領域の境界を定義する際, 境界のパラメータ表示を用いるが, そのパラメータを領域を左手に見て進むようにとらなくてはならない.




図 2.15:  $W$  型領域  $\Omega$  と境界ラベル

<sup>5</sup><http://www.im.uec.ac.jp/~koyama/w.html> からダウンロードできる.




図 2.16: W 型領域の三角形分割

— w-triangulation.edp —

```

int n=5;

border Gamma1(t = 0, 1) {x = -t -3; y = 4;}
border Gamma2(t = 0, 1) {x = -t +4; y = 4;}
border Gamma3(t = 0, 1) {x = 3*t -4; y = -6*t + 4;}
border Gamma4(t = 0, 1) {x = 3*t +1; y = 6*t - 2;}
border Gamma5(t = 0, 1) {x = -2*t -1; y = 4*t;}
border Gamma6(t = 0, 1) {x = -2*t +3; y = -4*t + 4;}
border Gamma7(t = 0, 1) {x = -t; y = -2*t + 2;}
border Gamma8(t = 0, 1) {x = -t+1; y = 2*t;}
border Gamma9(t = 0, 1) {x = t-1; y = 2*t-2;}
border Gamma10(t = 0, 1) {x = t; y = -2*t;}

mesh Th = buildmesh(Gamma1(n)+Gamma2(n)+Gamma3(6*n)+Gamma4(6*n)
                    +Gamma5(4*n)+Gamma6(4*n)+Gamma7(2*n)+Gamma8(2*n)
                    +Gamma9(2*n)+Gamma10(2*n));

plot(Th, wait=true, ps="w-triangulation.eps");

savemesh(Th, "w-triangulation.msh");

```

### 2.3.4 三角形分割データファイル

図 2.16 の三角形分割 Th のデータは，savemesh によって，ファイル w-triangulation.msh に保存される．ファイルに書かれるデータフォーマットは 4 つのデータ群からなる．第 1 データ群は，節点数，要素数，境界上の辺の数であり，第 2 データ群は「節点・座標対応表」に，第 3 データ群は「要素・節点对応表」に，第 4 データ群は「境界要素・節点对応表」に，それぞれ対応するデータである．これらのデータ群は表 2.6 のように並べられて出力される．

w-triangulation.msh の中身

```

382 612 150
-4 4 3
-3.79665632248 4 1
-3.90262050629 3.80524101257 3
-3.69793953756 3.80803382295 0
. . .
. . .
. . .

231 244 232 0
221 235 233 0
244 245 232 0
. . .
. . .
. . .

218 231 10
231 229 10
229 230 10
. . .
. . .
. . .

```

表 2.6: 三角形分割データ フォーマット

第 1 カラムの数	第 2 カラムの数	第 3 カラムの数	第 4 カラムの数	行数
節点数	要素数	境界上の辺の数	-	1 行
節点 $x$ 座標	節点 $y$ 座標	境界ラベル	-	節点数
要素の第 1 節点番号	要素の第 2 節点番号	要素の第 3 節点番号	部分領域ラベル	要素数
境界にある辺の第 1 節点番号	境界にある辺の第 2 節点番号	境界ラベル	-	境界上の辺の数

### 2.3.5 Poisson 方程式 (Laplace 方程式)

図 2.15 の W 型領域  $\Omega$  において次の Laplace 方程式の混合境界値問題を考える :

$$(P) \begin{cases} -\Delta u = 0 & \text{in } \Omega \\ u = 1 & \text{on } \Gamma_1, \\ u = 0 & \text{on } \Gamma_2, \\ \frac{\partial u}{\partial n} = 0 & \text{on } \Gamma_1, \Gamma_2 \text{ 以外の境界.} \end{cases}$$

この弱形式は

$$(II) \begin{cases} \text{Find } u \in V(g) \text{ such that} \\ \int_{\Omega} \left( \frac{\partial u}{\partial x} \frac{\partial v}{\partial x} + \frac{\partial u}{\partial y} \frac{\partial v}{\partial y} \right) dx dy = 0 \quad \forall v \in V. \end{cases}$$

ここで ,

$$(2.50) \quad \begin{aligned} V(g) &:= \{w \in H^1(\Omega) \mid w = g \text{ on } \Gamma_1 \cup \Gamma_2\}, \\ g &:= \begin{cases} 1 & \text{on } \Gamma_1, \\ 0 & \text{on } \Gamma_2, \end{cases} \\ V &:= \{v \in H^1(\Omega) \mid v = 0 \text{ on } \Gamma_1 \cup \Gamma_2\}. \end{aligned}$$

この問題を 2.3.3 節の三角形分割を用いて解くソースプログラム (w-laplace.edp) は以下のものである . ただし , 分割はより細かくしてある . すなわち ,  $n=10$  としてある .

```

int n=10;

border Gamma1(t = 0, 1) {x = -t -3; y = 4;}
border Gamma2(t = 0, 1) {x = -t +4; y = 4;}
border Gamma3(t = 0, 1) {x = 3*t -4; y = -6*t + 4;}
border Gamma4(t = 0, 1) {x = 3*t +1; y = 6*t - 2;}
border Gamma5(t = 0, 1) {x = -2*t -1; y = 4*t;}
border Gamma6(t = 0, 1) {x = -2*t +3; y = -4*t + 4;}
border Gamma7(t = 0, 1) {x = -t; y = -2*t + 2;}
border Gamma8(t = 0, 1) {x = -t+1; y = 2*t;}
border Gamma9(t = 0, 1) {x = t-1; y = 2*t-2;}
border Gamma10(t = 0, 1) {x = t; y = -2*t;}

mesh Th = buildmesh(Gamma1(n)+Gamma2(n)+Gamma3(6*n)+Gamma4(6*n)
                    +Gamma5(4*n)+Gamma6(4*n)+Gamma7(2*n)+Gamma8(2*n)
                    +Gamma9(2*n)+Gamma10(2*n));

fespace Vh(Th, P1);
Vh u, v;

solve laplace(u, v) = int2d(Th)(dx(u)*dx(v) + dy(u)*dy(v))
+ on(Gamma1, u=1) + on(Gamma2, u=0);

plot(u, wait=true, value=true, fill=true, ps="w-laplace.eps");

```

### 2.3.6 熱伝導方程式

有界領域  $\Omega$  において次の熱伝導方程式の初期値・境界値問題を考える：

$$(P) \begin{cases} \frac{\partial u}{\partial t}(\mathbf{x}, t) - \Delta u(\mathbf{x}, t) = f(\mathbf{x}, t) & \text{in } \Omega \times (0, T], \\ u(\mathbf{x}, t) = g(\mathbf{x}, t) & \text{on } \Gamma_D \times (0, T], \\ \frac{\partial u}{\partial n}(\mathbf{x}, t) = 0 & \text{on } \Gamma_N \times (0, T], \\ u(\mathbf{x}, 0) = u^0(\mathbf{x}) & \text{in } \Omega. \end{cases}$$

ここで，領域  $\Omega$  の境界  $\partial\Omega$  は二つの部分  $\Gamma_D$  と  $\Gamma_N$  からなるものとする．

この弱形式は

$$(II) \begin{cases} \text{Find } u : [0, T] \longrightarrow H^1(\Omega) \text{ such that} \\ \frac{d}{dt}(u(t), v) + a(u(t), v) = (f(t), v) \quad \forall v \in V, \\ u(t) = g(t) \quad \text{on } \Gamma_D, \\ u(0) = u^0 \quad \text{in } \Omega. \end{cases}$$




図 2.17: w-laplace.edp の出力結果

ここで,

$$\begin{aligned} (u(t), v) &:= \int_{\Omega} u(\mathbf{x}, t)v(\mathbf{x}) d\mathbf{x}, \\ a(u(t), v) &:= \int_{\Omega} \nabla u(\mathbf{x}, t) \cdot \nabla v(\mathbf{x}) d\mathbf{x}, \\ V &:= \{v \in H^1(\Omega) \mid v = 0 \text{ on } \Gamma_D\}. \end{aligned}$$

### 半離散近似問題

領域  $\Omega$  に三角形分割を施し, その節点を  $q^1, q^2, \dots, q^N$  とする. ただし, 記述を簡単にするために次のように番号付けされているものとする.

- $q^1, \dots, q^{N_0}$ :  $\Omega$  内部節点,
- $q^{N_0+1}, \dots, q^{N_0+N_2}$ :  $\Gamma_N$  内部節点,
- $q^{N'+1}, \dots, q^{N'+N_1}$ :  $\overline{\Gamma_D}$  上節点.

ここで,  $N' := N_0 + N_2$  とし,  $N = N' + N_1$  となるものとした.

節点  $q^i$  に対応する基底関数を  $\varphi_i$  とする. すなわち,  $\varphi_i$  は,  $\varphi_i(q^j) = \delta_{ij}$  ( $1 \leq i, j \leq N$ ) を満たす区分 1 次連続関数とする. この時, 関数空間 (有限要素空間):

$$\begin{aligned} V_h &:= \text{span}\{\varphi_i \mid 1 \leq i \leq N'\}, \\ W_h &:= \text{span}\{\varphi_i \mid 1 \leq i \leq N\} \end{aligned}$$

を導入する<sup>6</sup>.

<sup>6</sup> $V_h = \{v_h \in W_h \mid v_h = 0 \text{ on } \Gamma_D\}$  が成り立つことに注意する.

この時，弱形式  $(\Pi)$  の半離散近似問題を考えることができる：

$$(\Pi_h) \left\{ \begin{array}{l} \text{Find } u_h : [0, T] \longrightarrow W_h \text{ such that} \\ \frac{d}{dt}(u_h(t), v_h) + a(u_h(t), v_h) = (f(t), v_h) \quad \forall v_h \in V_h \\ u_h(t) = g_h(t) \quad \text{on } \Gamma_D, \\ u_h(0) = u_h^0 \quad \text{in } \Omega \end{array} \right.$$

ここで， $g_h$  および  $u_h^0$  はそれぞれ  $g$  および  $u^0$  の適当な近似関数である．

今，半離散近似問題  $(\Pi_h)$  における  $g_h$  を

$$g_h(\mathbf{x}, t) = \sum_{j=N'+1}^{N'+N_1} g(\mathbf{q}^j, t) \varphi_j(\mathbf{x})$$

で与えるものとする，問題  $(\Pi_h)$  の解  $u_h$  は

$$u_h(\mathbf{x}, t) = \sum_{j=1}^{N'} c_j(t) \varphi_j(\mathbf{x}) + \sum_{j=N'+1}^{N'+N_1} g(\mathbf{q}^j, t) \varphi_j(\mathbf{x})$$

と書ける．ただし， $c_j(t)$  ( $1 \leq j \leq N'$ ) は未知関数である．これを  $(\Pi_h)$  に代入し， $(\Pi_h)$  における  $v_h$  を  $\varphi_i$  ( $1 \leq i \leq N'$ ) とすると， $(\Pi_h)$  は次のように書ける：

$$\sum_{j=1}^{N'} \left[ b_{ij} \frac{dc_j}{dt}(t) + a_{ij} c_j(t) \right] = f_i(t) - \sum_{j=N'+1}^{N'+N_1} \left[ b_{ij} \frac{\partial g}{\partial t}(\mathbf{q}^j, t) + a_{ij} g(\mathbf{q}^j, t) \right] \quad (1 \leq i \leq N').$$

ここで，

$$\begin{aligned} a_{ij} &:= a(\varphi_j, \varphi_i) \quad (1 \leq i, j \leq N), \\ b_{ij} &:= (\varphi_j, \varphi_i) \quad (1 \leq i, j \leq N), \\ f_i(t) &:= (f(t), \varphi_i) \quad (1 \leq i \leq N) \end{aligned}$$

である．

さらに，

$$\begin{aligned} A &:= (a_{ij})_{1 \leq i, j \leq N'}, \\ B &:= (b_{ij})_{1 \leq i, j \leq N'}, \\ \mathbf{f}(t) &:= \left( f_i(t) - \sum_{j=N'+1}^{N'+N_1} \left[ b_{ij} \frac{\partial g}{\partial t}(\mathbf{q}^j, t) + a_{ij} g(\mathbf{q}^j, t) \right] \right)_{1 \leq i \leq N'}, \\ \mathbf{c}(t) &:= (c_i(t))_{1 \leq i \leq N'} \end{aligned}$$

とすると， $(\Pi_h)$  は次のように書ける：

$$(S_h) \left\{ \begin{array}{l} \text{Find } \mathbf{c} : [0, T] \longrightarrow \mathbb{R}^{N'} \text{ such that} \\ B \frac{d\mathbf{c}}{dt}(t) + A\mathbf{c}(t) = \mathbf{f}(t), \\ \mathbf{c}(0) = \mathbf{c}^0. \end{array} \right.$$

ここで,

$$u_h^0(\mathbf{x}) = \sum_{j=1}^{N'} c_j^0 \varphi_j(\mathbf{x}) + \sum_{j=N'+1}^{N'+N_1} g(\mathbf{q}^j, 0) \varphi_j(\mathbf{x})$$

$$\mathbf{c}^0 := (c_i^0)_{1 \leq i \leq N'}$$

とした.

### 全離散近似問題

時間微分  $d/dt$  を差分近似する. ここでは, 後退 Euler 法によって差分近似することを考える. 時間刻幅  $\tau$  として,  $t_n := n\tau$  ( $n = 0, 1, 2, \dots$ ) とする. そして,  $\mathbf{c}(t_n) \approx \mathbf{c}^n$ ,  $\mathbf{f}^n := \mathbf{f}(t_n)$  とする. この時,  $(S_h)$  の近似問題は次のようになる:

$$(S_h^\tau) \left\{ \begin{array}{l} \text{For each } n = 1, 2, \dots, \text{ find } \mathbf{c}^n \in \mathbb{R}^{N'} \text{ such that} \\ B \left( \frac{\mathbf{c}^{n+1} - \mathbf{c}^n}{\tau} \right) + A\mathbf{c}^{n+1} = \mathbf{f}^{n+1}. \end{array} \right.$$

問題  $(S_h^\tau)$  の等式は,

$$(2.51) \quad (B + \tau A)\mathbf{c}^{n+1} = B\mathbf{c}^n + \tau \mathbf{f}^{n+1}$$

と書ける.

同様の操作を,  $(\Pi_h)$  に対して行うことを考える.  $u_h(t_n) \approx u_h^n$  とする. この時,  $(\Pi_h)$  の近似問題は次のようになる:

$$(\Pi_h^\tau) \left\{ \begin{array}{l} \text{For each } n = 1, 2, \dots, \text{ find } u_h^n \in W_h \text{ such that} \\ \left( \frac{u_h^{n+1} - u_h^n}{\tau}, v_h \right) + a(u_h^{n+1}, v_h) = (f(t_{n+1}), v_h) \quad \forall v_h \in V_h \\ u_h^{n+1} = g_h(t_{n+1}) \quad \text{on } \Gamma_D, \\ u_h(0) = u_h^0 \quad \text{in } \Omega \end{array} \right.$$

問題  $(\Pi_h^\tau)$  の等式は,

$$\int_{\Omega} u_h^{n+1} v_h \, dx dy + \int_{\Omega} \tau \left( \frac{\partial u_h^{n+1}}{\partial x} \frac{\partial v_h}{\partial x} + \frac{\partial u_h^{n+1}}{\partial y} \frac{\partial v_h}{\partial y} \right) dx dy - \int_{\Omega} u_h^n v_h \, dx dy + \tau \int_{\Omega} f(t_{n+1}) v_h \, dx dy = 0 \quad \forall v_h \in V_h$$

と書ける. FreeFEM++ では, この書き方を利用する.

問題  $(P)$  において,  $\Omega$  を図 2.15 の W 型領域とし,  $\Gamma_D$  を図 2.15 の  $\Gamma_1 \cup \Gamma_2$  とし,  $\Gamma_N$  を残りの境界とする.  $f \equiv 0$  とし,  $g$  は 2.3.5 節の (2.50) とし,  $u^0 \equiv 0$  とする. 近似問題  $(\Pi_h^\tau)$  を解くソースプログラム (w-heat.edp) は以下のものである.

註記 2.12 実際, 自分でプログラムを作成する際には,

$$(S_h) \left\{ \begin{array}{l} \text{Find } \begin{bmatrix} \mathbf{c} \\ \mathbf{c}_D \end{bmatrix} : [0, T] \longrightarrow \mathbb{R}^N \text{ such that} \\ \begin{bmatrix} B & O \\ O & O \end{bmatrix} \frac{d}{dt} \begin{bmatrix} \mathbf{c}(t) \\ \mathbf{c}_D(t) \end{bmatrix} + \begin{bmatrix} A & O \\ O & I \end{bmatrix} \begin{bmatrix} \mathbf{c}(t) \\ \mathbf{c}_D(t) \end{bmatrix} = \begin{bmatrix} \mathbf{f}(t) \\ \mathbf{g}(t) \end{bmatrix}, \\ \begin{bmatrix} \mathbf{c}(0) \\ \mathbf{c}_D(0) \end{bmatrix} = \begin{bmatrix} \mathbf{c}^0 \\ \mathbf{g}(0) \end{bmatrix}. \end{array} \right.$$

を考える．ここで，

$$\mathbf{g}(t) := (g(\mathbf{q}^j, t))_{N'+1 \leq j \leq N'+N}$$

ある．すなわち，行列は  $N \times N$  のサイズで考え，Dirichlet 条件の処理を施す．この時，(2.51) は

$$\begin{bmatrix} B + \tau A & O \\ O & \tau I \end{bmatrix} \begin{bmatrix} \mathbf{c}^{n+1} \\ \mathbf{c}_D^{n+1} \end{bmatrix} = \begin{bmatrix} B & O \\ O & O \end{bmatrix} \begin{bmatrix} \mathbf{c}^n \\ \mathbf{c}_D^n \end{bmatrix} + \tau \begin{bmatrix} \mathbf{f}^{n+1} \\ \mathbf{g}^{n+1} \end{bmatrix}$$

となる．ここで， $\mathbf{g}^n := \mathbf{g}(t_n)$  である．

w-heat.edp

```
int n=10;
real T = 40, tau = 0.1;

border Gamma1(t = 0, 1) {x = -t -3; y = 4;}
border Gamma2(t = 0, 1) {x = -t +4; y = 4;}
border Gamma3(t = 0, 1) {x = 3*t -4; y = -6*t + 4;}
border Gamma4(t = 0, 1) {x = 3*t +1; y = 6*t - 2;}
border Gamma5(t = 0, 1) {x = -2*t -1; y = 4*t;}
border Gamma6(t = 0, 1) {x = -2*t +3; y = -4*t + 4;}
border Gamma7(t = 0, 1) {x = -t; y = -2*t + 2;}
border Gamma8(t = 0, 1) {x = -t+1; y = 2*t;}
border Gamma9(t = 0, 1) {x = t-1; y = 2*t-2;}
border Gamma10(t = 0, 1) {x = t; y = -2*t;}

mesh Th = buildmesh(Gamma1(n)+Gamma2(n)+Gamma3(6*n)+Gamma4(6*n)
                    +Gamma5(4*n)+Gamma6(4*n)+Gamma7(2*n)+Gamma8(2*n)
                    +Gamma9(2*n)+Gamma10(2*n));

fespace Vh(Th, P1);
Vh u=0, v, uold;

problem heat(u, v) = int2d(Th)(u*v + tau*(dx(u)*dx(v) + dy(u)*dy(v)))
                    - int2d(Th)(uold*v)
                    + on(Gamma1, u=1) + on(Gamma2, u=0);

for(real t=0; t<T; t+=tau){
  uold = u;
  heat;
  plot(u, fill=true);
}
```

### 2.3.7 Stokes 方程式

有界領域  $\Omega \subset \mathbb{R}^2$  において次の Stokes 方程式<sup>7</sup>の境界値問題を考える：

$$(P) \begin{cases} -\Delta \mathbf{u} + \nabla p = \mathbf{f} & \text{in } \Omega, \\ \operatorname{div} \mathbf{u} = 0 & \text{in } \Omega, \\ \mathbf{u}(\mathbf{x}, t) = \mathbf{g} & \text{on } \Gamma_D, \\ \frac{\partial \mathbf{u}}{\partial \mathbf{n}} - p \mathbf{n} = \mathbf{t} & \text{on } \Gamma_N. \end{cases}$$

ここで，領域  $\Omega$  の境界  $\partial\Omega$  は二つの部分  $\Gamma_D$  と  $\Gamma_N$  からなるものとし， $\mathbf{n} := (n_1, n_2)$  は外向き単位法線ベクトルである．問題 (P) は，外力  $\mathbf{f} := (f_1, f_2)$ ，境界での速度  $\mathbf{g} := (g_1, g_2)$ ，境界での応力の法線方向成分  $\mathbf{t} := (t_1, t_2)$  が既知の時，流体の流速  $\mathbf{u} := (u_1, u_2)$  と圧力  $p$  を求める問題である．

この弱形式は

$$(II) \begin{cases} \text{Find } \{\mathbf{u}, p\} \in V(\mathbf{g}) \times Q \text{ such that} \\ \int_{\Omega} \nabla \mathbf{u} : \nabla \mathbf{v} \, d\mathbf{x} - \int_{\Omega} p \operatorname{div} \mathbf{v} \, d\mathbf{x} = \int_{\Omega} \mathbf{f} \cdot \mathbf{v} \, d\mathbf{x} + \int_{\Gamma_N} \mathbf{t} \cdot \mathbf{v} \, d\Gamma_N & \forall \mathbf{v} \in V, \\ \int_{\Omega} q \operatorname{div} \mathbf{u} \, d\mathbf{x} = 0 & \forall q \in Q. \end{cases}$$

ここで，

$$\begin{aligned} \nabla \mathbf{u} : \nabla \mathbf{v} &:= \sum_{i=1}^2 \nabla u_i \cdot \nabla v_i, \\ V(\mathbf{g}) &:= \left\{ \mathbf{w} \in [H^1(\Omega)]^2 \mid \mathbf{w} = \mathbf{g} \text{ on } \Gamma_D \right\}, \\ V &:= \left\{ \mathbf{v} \in [H^1(\Omega)]^2 \mid \mathbf{v} = 0 \text{ on } \Gamma_D \right\}, \\ Q &:= \left\{ q \in L^2(\Omega) \mid \int_{\Omega} q \, d\mathbf{x} = 0 \right\}. \end{aligned}$$

領域  $\Omega$  を図 2.18 のような流入口付き W 型領域とする．境界に図 2.18 のようにラベルをつける．境界  $\Gamma_N := \Gamma_2$ ，境界  $\Gamma_D$  をその他の部分とする．既知データは  $\mathbf{f} = \mathbf{o}$ ， $\mathbf{t} = \mathbf{o}$ ，

$$\mathbf{g} = \begin{cases} \left( -\frac{a}{4}(y-4)(y-5), 0 \right) & \text{on } \Gamma_{11} \text{ (流入口ではポアズイコ流れとする)} \\ \mathbf{o} & \text{on } \Gamma_D \text{ のその他の部分} \end{cases}$$

とする．

この時，(II) の二つの等式はまとめて，次のように書けることに注意する：

$$\begin{aligned} & \int_{\Omega} \left( \frac{\partial u_1}{\partial x} \frac{\partial v_1}{\partial x} + \frac{\partial u_1}{\partial y} \frac{\partial v_1}{\partial y} + \frac{\partial u_2}{\partial x} \frac{\partial v_2}{\partial x} + \frac{\partial u_2}{\partial y} \frac{\partial v_2}{\partial y} \right) dx dy \\ & - \int_{\Omega} p \left( \frac{\partial v_1}{\partial x} + \frac{\partial v_2}{\partial y} \right) dx dy + \int_{\Omega} q \left( \frac{\partial u_1}{\partial x} + \frac{\partial u_2}{\partial y} \right) dx dy = 0 \quad \forall (v_1, v_2, q) \in V \times Q. \end{aligned}$$

FreeFEM++ ではこの書き方を用いる．

註記 2.13 Stokes 問題の有限要素計算では， $V$  と  $Q$  を近似する有限要素空間  $V_h$  と  $Q_h$  を適切な組み合わせで選ばなくてはならない．

<sup>7</sup>Stokes 方程式は流速の遅い非圧縮粘性流体の運動を記述する方程式である．




図 2.18: 流入口付き W 型領域と境界ラベル

$V_h$ :P2,  $Q_h$ :P1

×  $V_h$ :P1,  $Q_h$ :P1

これらの適切性を判断するための条件として, Inf-Sup 条件がある:  $\exists \beta > 0$  such that  $\forall h$

$$\inf_{q_h \in Q_h \setminus \{0\}} \sup_{\mathbf{v}_h \in V_h \setminus \{0\}} \frac{\int_{\Omega} q_h \operatorname{div} \mathbf{v}_h \, d\mathbf{x}}{\|\mathbf{v}_h\|_V \|q_h\|_Q} \geq \beta.$$

ここで,  $\beta$  は  $h$  に依らない定数である.

註記 2.14 FreeFEM++ (w-stokes.edp) では,  $Q_h$  の元  $q$  は,

$$\int_{\Omega} q \, dx dy = 0$$

をみたとように設定されていない. このことから, 解くべき連立 1 次方程式の行列のランクは 1 つ下がる (ものと思われる). 実際, solver=LU とするとエラーになる.

```

int n=5;
real a=10;
func ud = -a*0.25*(y-4)*(y-5);

border Gamma1(t = 0, 1) {x = -1.5*t -3.5; y = 5;}
border Gamma2(t = 0, 1) {x = -t +4; y = 4;}
border Gamma3(t = 0, 1) {x = 3*t -4; y = -6*t + 4;}
border Gamma4(t = 0, 1) {x = 3*t +1; y = 6*t - 2;}
border Gamma5(t = 0, 1) {x = -2.5*t -1; y = 5*t;}
border Gamma6(t = 0, 1) {x = -2*t +3; y = -4*t + 4;}
border Gamma7(t = 0, 1) {x = -t; y = -2*t + 2;}
border Gamma8(t = 0, 1) {x = -t+1; y = 2*t;}
border Gamma9(t = 0, 1) {x = t-1; y = 2*t-2;}
border Gamma10(t = 0, 1) {x = t; y = -2*t;}
border Gamma11(t = 0, 1) {x = -5; y = -t+5;}
border Gamma12(t = 0, 1) {x = t-5; y = 4;}

mesh Th = buildmesh(Gamma1(1.5*n)+Gamma2(n)+Gamma3(6*n)+Gamma4(6*n)
                    +Gamma5(5*n)+Gamma6(4*n)+Gamma7(2*n)+Gamma8(2*n)
                    +Gamma9(2*n)+Gamma10(2*n)+Gamma11(n)+Gamma12(n));

fespace Vh(Th,P2); Vh u1,u2,v1,v2;
fespace Qh(Th,P1); Qh p,q;

solve stokes([u1,u2,p],[v1,v2,q],solver=UMFPACK) =
  int2d(Th)(dx(u1)*dx(v1)+dy(u1)*dy(v1) + dx(u2)*dx(v2)+ dy(u2)*dy(v2)
    - p*(dx(v1)+dy(v2)) + q*(dx(u1)+dy(u2)))
  + on(1,3,4,5,6,7,8,9,10,12,u1=0,u2=0) + on(11,u1=ud,u2=0);

plot([u1,u2],p,wait=1,ps="w-stokes.eps");
//plot(Th, wait=true, ps="w-stokes-Th.eps");
//savemesh(Th, "w-stokes-Th.msh");

```




図 2.19: w-stokes.edp の出力結果